PRÁCTICA 3 - RESOLUCIÓN

Interrupciones por Hardware

Objetivos: Comprender la utilidad de las interrupciones por hardware y el funcionamiento del Controlador de Interrupciones Programable (PIC). Utilizar el TIMER, F10, y el dispositivo de handshaking (HAND-SHAKE) mediante interrupciones para el intercambio de información entre el microprocesador y el mundo exterior.

Parte 1: PIC y Vector de Interrupciones

1) **Registros del PIC** ightharpoonup Dados los siguientes registros del PIC indique su función. Además, para el valor de ejemplo, indique qué significa en términos del funcionamiento del programa

Dirección	Registro	Función	Valor de Ejemplo
20h	EOI	Fin de Interrupción. Solo para escribir, se debe mandar el comando de fin de interrupción cuando se terminó de servir la interrupción. (Valor 20h)	No aplica
21h	IMR	Registro de Mascara de Interrupciones. Permite enmascarar selectivamente las interrupciones que va a recibir el PIC. Cada bit de este registro representa una línea de interrupción (bit 0 se asocia a la entrada INT0 bit 7 se asocia a la entrada INT7). Si el valor es puesto en 1, esa interrupción estará inhibida.	1111 1101 Esta habilitada la interrupción INT1 (bit 1 en 0). Esa interrupción está conectada al Timer.
22h	IRR	Registro de Petición de Interrupciones. Almacena las interrupciones pendientes. Cada bit de este registro representa una línea de interrupción (bit 0 se asocia a la entrada INTO bit 7 se asocia a la entrada INT7). Cuando la interrupción se satisface, el bit de dicha línea se pone a cero.	0000 0011 Se recibió y están pendientes de ser atendidas las interrupciones INTO e INT1. Esa interrupción está conectada a la tecla F-10 y al Timer.

23h	ISR	Registro de Interrupción en Servicio. El bit que representa la línea de interrupción que se está atendiendo estará en 1 interrupción (bit 0 se asocia a la entrada INTO bit 7 se asocia a la entrada INT7), el resto en 0. Si no hay interrupciones atendiéndose todos los bits valen 0.	0000 0100 La CPU está ejecutando el manejador de interrupción INT2. Esa interrupción está conectada al HandShake.
24h	INT0	ID de Línea INTO. Almacena el ID de la interrupción asociada al dispositivo F10 para buscar en el vector de interrupciones la dirección de comienzo de la subrutina que lo atiende.	Es el número de vector que se le enviará a la CPU para manejar la interrupción. Si multiplico por 4 (15×4=60) nos da la dirección de memoria donde debe estar la dirección de rutina que maneja la interrupción correspondiente a F10.
25h	INT1	ID de Línea INT1. Almacena el ID de la interrupción asociada al dispositivo Timer para buscar en el vector de interrupciones la dirección de comienzo de la subrutina que lo atiende.	Es el número de vector que se le enviará a la CPU para manejar la interrupción. Si multiplico por 4 (10×4=40) nos da la dirección de memoria donde debe estar la dirección de rutina que maneja la interrupción correspondiente al <i>Timer</i> .
26h	INT2	ID de Línea INT2. Almacena el ID de la interrupción asociada al dispositivo Handshake para buscar en el vector de interrupciones la dirección de comienzo de la subrutina que lo atiende.	Es el número de vector que se le enviará a la CPU para manejar la interrupción. Si multiplico por 4 (25×4=100) nos da la dirección de memoria donde debe estar la dirección de rutina que maneja la interrupción correspondiente al Handshake.

2) **Configuración del PIC** Y Se desean configurar algunas interrupciones mediante el PIC. Complete la tabla indicando en cada caso los registros o direcciones que faltan para que funcionen las interrupciones, o el dispositivo que se está usando.

Dispositivos	IMR	INT0	INT1	INT2	Dirección Vector
F10	1111 1110	5	No importa	No importa	20

F10	1111 1110	8	No importa	No importa	32
Timer	1111 1101	No importa	10	No importa	40
HANDSHAKE	1111 1011	No importa	No importa	5	20
Sin interrupciones	1111 1111	No importa	No importa	No importa	-
F10 y Timer	1111 1100	10	20	No importa	40 y 80
TIMER y HANDSHAKE	1111 1001	No importa	4	8	16 y 32
F10, TIMER y HANDSHAKE	1111 1000	5	10	15	20, 40, 60

Nota: En el caso de haber más de un dispositivo en uso, indicarlos en el mismo orden en las columnas "Dispositivos" y "Dirección Vector". En el caso en que una columna no deba configurarse, indicarla como "No importa".

Parte 2: Interrupciones con la tecla F10, Timer y Handshake

3) **Contador de pulsaciones** $\not\leftarrow$ El siguiente programa cuenta el número de veces que se presiona la tecla F10 y acumula este valor en el registro DX. El programa nunca termina, ya que ejecuta un ciclo infinito. Completar las instrucciones faltantes para que el programa funcione correctamente.

EOI EQU 20H	ORG 2000H	ORG 3000H
IMR EQU 21H	CLI	RUT_F10: PUSH AX
<u>INTØ EQU 24h</u>	MOV AL, 0FEH	<u>INC DX</u>
	<u>OUT IMR, AL</u>	MOV AL, EOI
N_F10 EQU 15	MOV AL, N_F10	<u>OUT EOI, AL</u>
	OUT INT0, AL	POP AX
<u>ORG 60</u>	MOV DX, 0	IRET
IP_F10 DW RUT_F10	STI	END
	LAZO: JMP LAZO	

Luego de completarlo, verifique su funcionamiento en el simulador, y explicar detalladamente:

A. ¿Qué hacen las instrucciones CLI y STI? ¿Qué podría suceder si no están las mismas y la persona que usa el programa presiona F10 rápidamente durante la configuración del PIC?

La instrucción **CLI** inhibe las interrupciones enmascarables. la instrucción **STI** las vuelve a habilitar. Las interrupciones enmascarables son las interrupciones gestionadas por el PIC.

Si no están las mismas mientras configuramos el PIC, la CPU podría recibir un pedido de interrupción de un vector que aún no este configurado (en este caso el registro INTO), eso podría hacer que la CPU busque la rutina manejadora de la interrupción en la dirección equivocada y se termine ejecutando código en otra dirección de memoria.

B. ¿Por qué se usa el valor 0FEH en el programa?

El valor OFEH lo usamos para configurar el registro **IMR** del PIC, para habilitar solamente la interrupción **F10** del simulador.

C. ¿Qué instrucciones le indican al PIC que la interrupción ha terminado?

Se debe escribir el valor 20h en el registro EOI para indicar al PIC que se terminó de procesar la interrupción. Esto lo hacemos con las instrucciones:

MOV AL, EOI OUT EOI, AL

D. ¿En qué dirección se encuentra la subrutina que atiende las interrupciones del F10?

La subrutina que atiende la interrupción se encuentra a partir de la dirección 3000h.

4) Tecla F10 con pantalla y terminación 🛧 🛧

A) Escribir un programa que muestre en pantalla "Vamos las interrupciones!" cada vez que se presiona la tecla F10. El programa nunca termina.

```
EOI EQU 20H
IMR EQU 21H
INTO EQU 24h
N_F10
        EQU 15
ORG 60
IP_F10 DW RUT_F10
 ORG 1000H
          DB "Vamos las interrupciones!"
TEXT0
FIN_TEXTO DB ?
 ORG 3000H
RUT_F10: PUSH AX
         PUSH BX
         MOV BX, OFFSET TEXTO
         MOV AL, OFFSET FIN_TEXTO - OFFSET TEXTO
         INT 7
         MOV AL, EOI
         OUT EOI, AL
         POP BX
         POP AX
         IRET
      ORG 2000H
      CLI
      MOV AL, ØFEH
      OUT IMR, AL
      MOV AL, N_F10
      OUT INTO, AL
      STI
LAZO: JMP LAZO
END
```

B) Idem A pero ahora el programa termina luego de mostrar 5 veces el mensaje (es decir, luego de que presione 5 veces la tecla F10).

Pista: cuando se ejecuta la 5ta interrupción, deben deshabilitarse las mismas para evitar reaccionar a próximas pulsaciones de la tecla F10. Utilizar una variable global para registrar cuántas veces falta mostrar el mensaje antes de terminar.

```
EOI EQU 20H
IMR EQU 21H
INTO EQU 24h
 ORG 1000H
          DB "Vamos las interrupciones!"
TEXTO
FIN TEXTO DB ?
          DB 0
CANT
ORG 2000H
      CLI
      MOV AL, 0FEH
      OUT IMR, AL
      MOV AL, N_F10
      OUT INTO, AL
      STI
LAZO: CMP CANT, 5
      JNZ LAZO
      INT 0
 ORG 3000H
RUT_F10: PUSH AX
         PUSH BX
         MOV BX, OFFSET TEXTO
         MOV AL, OFFSET FIN_TEXTO - OFFSET TEXTO
         INT 7
         INC CANT
         CMP CANT, 5
         JNZ SIGUE
         MOV AL, 0FFh
         OUT IMR, AL
SIGUE:
         MOV AL, EOI
         OUT EOI, AL
         POP BX
         POP AX
         IRET
END
```

5) Timer, tres usos: periódico infinito, periódico finito, y única vez 🜟 🜟

 Escribir un programa que muestra el mensaje "Vamos las interrupciones!" una vez cada 2 segundos y no termine nunca.

```
EOI EQU 20H
IMR EQU 21H
INT1 EQU 25h

CONT EQU 10h
COMP EQU 11h

N_TIMER EQU 20
```

```
ORG 80
IP_F10 DW RUT_TIMER
 ORG 1000H
TEXTO DB "Vamos las interrupciones!"
FIN_TEXTO DB ?
 ORG 2000H
     CLI
     MOV AL, ØFDH
     OUT IMR, AL
     MOV AL, N_TIMER
     OUT INT1, AL
     MOV AL, 2
     OUT COMP, AL
     MOV AL, 0
     OUT CONT, AL
     STI
LAZO: JMP LAZO
 ORG 3000H
RUT_TIMER: PUSH AX
     PUSH BX
     MOV BX, OFFSET TEXTO
     MOV AL, OFFSET FIN_TEXTO - OFFSET TEXTO
     INT 7
     MOV AL, EOI
           OUT EOI, AL
     MOV AL, 0
     OUT CONT, AL
     POP BX
           POP AX
           IRET
END
```

b) Modificar a) para que termine a los 10 segundos.

```
EOI EQU 20H
IMR EQU 21H
INT1 EQU 25h

CONT EQU 10h
COMP EQU 11h

N_TIMER EQU 20

ORG 80
IP_F10 DW RUT_TIMER

ORG 1000H
TEXTO DB "Vamos las interrupciones!"
FIN_TEXTO DB ?
CANT DB 0

ORG 2000H
CLI
```

```
MOV AL, 0FDH
     OUT IMR, AL
     MOV AL, N_TIMER
     OUT INT1, AL
     MOV AL, 2
 OUT COMP, AL
 MOV AL, 0
 OUT CONT, AL
 STI
LAZO: CMP CANT, 5
     JNZ LAZO
     INT 0
 ORG 3000H
RUT_TIMER: PUSH AX
     PUSH BX
     MOV BX, OFFSET TEXTO
     MOV AL, OFFSET FIN_TEXTO - OFFSET TEXTO
     INT 7
           INC CANT
           CMP CANT, 5
           JNZ SIGUE
           MOV AL, 0FFh
           OUT IMR, AL
SIGUE:
           MOV AL, EOI
           OUT EOI, AL
     MOV AL, 0
     OUT CONT, AL
     POP BX
           POP AX
           IRET
END
```

c) Modificar a) para que solo imprima una vez, a los 10 segundos, y luego termine.

```
EOI EQU 20H
IMR EQU 21H
INT1 EQU 25h

CONT EQU 10h
COMP EQU 11h

N_TIMER EQU 20

ORG 80
IP_F10 DW RUT_TIMER

ORG 1000H
TEXTO DB "Vamos las interrupciones!"
FIN_TEXTO DB ?
TERMINA DB 0

ORG 3000H
```

```
RUT_TIMER: PUSH AX
     PUSH BX
     MOV BX, OFFSET TEXTO
     MOV AL, OFFSET FIN_TEXTO - OFFSET TEXTO
     INT 7
            INC TERMINA
           MOV AL, 0FFh
            OUT IMR, AL
SIGUE:
           MOV AL, EOI
           OUT EOI, AL
     MOV AL, 0
     OUT CONT, AL
     POP BX
            POP AX
            IRET
 ORG 2000H
     CLI
           MOV AL, 0FDH
            OUT IMR, AL
           MOV AL, N_TIMER
            OUT INT1, AL
            MOV AL, 10
            OUT COMP, AL
            MOV AL, 0
            OUT CONT, AL
            STI
            CMP TERMINA, 0
LAZO:
            JZ LAZO
            INT 0
END
```

d) **Desafío:** Modificar a) para que el primer mensaje se imprima luego de 1 segundo, el segundo luego de 2 segundos, el tercero luego de 3, y así sucesivamente, hasta que se espere 255 en el último mensaje, y el programa termine

```
EOI EQU 20H
IMR EQU 21H
INT1 EQU 25h
CONT EQU 10h
COMP EQU 11h
N_TIMER
          EQU 20
ORG 80
IP_F10 DW RUT_TIMER
 ORG 1000H
TEXTO DB "Vamos las interrupciones!"
FIN_TEXTO DB ?
ESPERA DB 1
 ORG 3000H
RUT_TIMER: PUSH AX
     PUSH BX
     MOV BX, OFFSET TEXTO
```

```
MOV AL, OFFSET FIN_TEXTO - OFFSET TEXTO
     INT 7
            INC ESPERA
            MOV AL, ESPERA
            OUT COMP, AL
            CMP AL, 0
            JNZ SIGUE
            MOV AL, 0FFh
            OUT IMR, AL
SIGUE:
            MOV AL, EOI
            OUT EOI, AL
     MOV AL, 0
     OUT CONT, AL
     POP BX
            POP AX
            IRET
   ORG 2000H
      CLI
      MOV AL, 0FDH
      OUT IMR, AL
      MOV AL, N_TIMER
      OUT INT1, AL
      MOV AL, ESPERA
      OUT COMP, AL
      MOV AL, 0
      OUT CONT, AL
  STI
LAZO: CMP ESPERA, 0
      JNZ LAZO
      INT 0
END
```

- 6) F10 con Timer
- a) Conteo regresivo ☆☆: Escribir un programa que implemente un conteo regresivo a partir de un valor (de 1 a 9) ingresado desde el teclado. El conteo debe comenzar al presionarse la tecla F10. El tiempo transcurrido debe mostrarse en pantalla, actualizándose el valor cada segundo. Por ejemplo, si se ingresa el valor 7, cuando se aprieta F10 debe mostrarse en pantalla "7 6 5 4 3 2 1 0" en los 7 segundos siguientes

Pista: Una vez leído el dígito se puede activar solamente la interrupción F10, y cuando se invoque la subrutina F10 activar el Timer para que a partir de ese momento se muestre el conteo regresivo.

```
CONT EQU 10H
COMP EQU 11H
PIC EQU 20H
EOI EQU 20H
IMR EQU 21H
INT0 EQU 24h
INT1 EQU 25H

N_F10 EQU 10
N_CLK EQU 15

ORG 40
```

```
IP_F10 DW RUT_F10
     ORG 60
IP_CLK DW
                RUT_CLK
     ORG 1000H
CONTADOR DB 39H
FINAL
           DB 0
     ORG 3000H
      RUT_CLK:
                 PUSH AX
           PUSH BX
                 MOV BX, OFFSET CONTADOR
                 MOV AL, 1
                 INT 7
                 CMP CONTADOR, 30h
                 JNZ SIGUE
                 MOV AL, 0FFH
                 OUT IMR, AL
                 MOV FINAL, 1
 SIGUE:
           DEC CONTADOR
                 MOV AL, 0
                 OUT CONT, AL
                 MOV AL, EOI
                 OUT EOI, AL
                 POP BX
                 POP AX
                 IRET
    ORG 3200H
      RUT_F10:
                 PUSH AX
                 MOV AL, 0FDH
                 OUT IMR, AL
                 MOV AL, 0
                 OUT CONT, AL
                 MOV AL, 1
                 OUT COMP, AL
                 MOV AL, EOI
                 OUT EOI, AL
                 POP AX
                 IRET
     ORG 3400H
   LEE_NUMERO:
                 MOV BX, OFFSET CONTADOR
       REPITE:
                 INT 6
                 MOV AL, CONTADOR
                 CMP AL, 30H
                 JS REPITE
           CMP AL, 39H
           JNS REPITE
                 RET
     ORG 2000H
           CALL LEE_NUMERO
           CLI
           MOV AL, ØFEH
```

```
OUT IMR, AL
MOV AL, N_F10
OUT INT0, AL
MOV AL, N_CLK
OUT INT1, AL
STI
BUCLE: CMP FINAL, 0
JZ BUCLE
INT 0
```

Pista: Utilizar un flag de estado "PAUSA" que se controla mediante la tecla F10, y que la subrutina del timer puede consultar para saber si debe decrementar el contador y mostrar en pantalla.

```
CONT EQU 10H
COMP EQU 11H
PIC
     EQU 20H
EOI
     EQU 20H
IMR
     EQU 21H
INT0
     EQU 24h
INT1 EQU 25H
N_F10 EQU 10
N CLK EQU 15
    ORG 40
IP_F10 DW RUT_F10
     ORG 60
IP_CLK DW RUT_CLK
     ORG 1000H
CONTADOR
                  DB 39H
                  DB 0
PAUSA
                  DB 0
TERMINO_CONTEO
     ORG 3000H
      RUT_CLK:
                  PUSH AX
                  PUSH BX
                  MOV BX, OFFSET CONTADOR
                  MOV AL, 1
                  INT 7
                  CMP CONTADOR, 30h
                  JNZ SIGUE
                  MOV AL, ØFEH
                  OUT IMR, AL
                  MOV TERMINO_CONTEO, 1
 SIGUE:
           DEC CONTADOR
                  MOV AL, 0
                  OUT CONT, AL
                  MOV AL, EOI
                  OUT EOI, AL
                  POP BX
```

```
POP AX
                  IRET
   ORG 3200H
      RUT_F10: PUSH AX
               CMP TERMINO_CONTEO, 1
               JNZ NO_REPETIR
               MOV TERMINO_CONTEO, 0
               MOV CONTADOR, 39H
               MOV AL, 0
               OUT CONT, AL
               MOV AL, 0FCh
               JMP CONT_F10
  NO_REPETIR: CMP PAUSA, 0
               JZ PAUSAR
               MOV PAUSA, 0
               MOV AL, ØFCH
               JMP CONT_F10
       PAUSAR: MOV PAUSA, 1
               MOV AL, 0
               OUT CONT, AL
               MOV AL, 0FEh
      CONT_F10:OUT IMR, AL
               MOV AL, EOI
               OUT EOI, AL
               POP AX
               IRET
 ORG 2000H
   CLI
   MOV AL, ØFCH
   OUT IMR, AL
   MOV AL, N_F10
   OUT INT0, AL
   MOV AL, N_CLK
   OUT INT1, AL
   MOV AL, 0
   OUT CONT, AL
       MOV AL, 1
   OUT COMP, AL
BUCLE: JMP BUCLE
END
```

) Impresión con HANDSHAKE mediante interrupciones 🛨 🛨

a) Escribir un programa que imprime "UNIVERSIDAD NACIONAL DE LA PLATA" en la impresora a través del HAND-SHAKE. El envío de los caracteres se realiza por **interrupciones** emitidas desde el HAND-SHAKE cada vez que detecta que la impresora se desocupa.

```
IMR EQU 21H
EOI EQU 20H
INT2 EQU 26H

HAND EQU 40H
EST EQU 41H
```

```
DATO EQU 40H
N_HND EQU 10
        ORG 40
        DW RUT_HND
        ORG 1000H
        DB "UNIVERSIDAD NACIONAL DE LA PLATA"
MSJ
FIN
        DB ?
        ORG 2000H
        MOV BX, OFFSET MSJ
        MOV CL, OFFSET FIN - OFFSET MSJ
        MOV AL, 0FBH
        OUT IMR, AL
        MOV AL, N_HND
        OUT INT2, AL
        MOV AL, 80H
        OUT EST, AL
        STI
 LAZO:
        CMP CL, 0
        JNZ LAZO
        IN AL, EST
        AND AL, 7FH
        OUT EST, AL
        INT 0
        ORG 3000H
RUT_HND: PUSH AX
         MOV AL, [BX]
         OUT DATO, AL
         INC BX
         DEC CL
         JNZ FINAL
         MOV AL, 0FFH
         OUT IMR, AL
  FINAL: MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
END
```

- b) Probar el programa utilizando:
 - 1) Una **velocidad de reloj** del 100%, y una **velocidad de impresión** del 25%
 - 2) Una **velocidad de reloj** del 25%, y una **velocidad de impresión** del 100% ¿En qué caso sería mejor utilizar consulta de estado en lugar de interrupciones?

En el caso de una CPU lenta y una impresora rápida (caso 2) la consulta de estado resulta más conveniente. La complejidad que añaden las interrupciones no se justifica si la espera es poca.

c) Modificar a) de modo que el string a imprimir se lea desde teclado. El string a leer tiene una longitud fija de 10 caracteres, y se lee de forma completa *antes* de comenzar la impresión. Tener en cuenta entonces que las interrupciones deben esperar a que se ingrese todo el string.

```
DATO EQU 40h
EST EQU 41h
IMR EQU 21h
INT2 EQU 26h
EOI EQU 20h
     ORG 40
     DW RUT_HAND
ORG 3000H
 RUT_HAND: MOV AL, [BX]
           OUT DATO, AL
     INC BX
     DEC CL
     JNZ FIN
     IN AL, EST
          AND AL, 07FH
     OUT EST, AL
           MOV AL, 0FFh
           OUT IMR, AL
 FIN:
           MOV AL, EOI
     OUT EOI, AL
     IRET
        ORG 1000H
NUM_CAR DB 5
CAR
     DB ?
           ORG 2000H
           ; PROGRAMA PRINCIPAL
           MOV BX, OFFSET CAR
           MOV CL, NUM_CAR
LAZO:
           INT 6
           INC BX
     DEC CL
     JNZ
         LAZ0
           MOV BX, OFFSET CAR
           MOV CL, NUM_CAR
     CLI
     MOV AL, 0FBh
           OUT IMR, AL
           MOV AL, 10
           OUT INT2, AL
           IN AL, EST
     OR AL, 080h
     OUT EST, AL
           STI
            CMP CL, 0
ESPERA:
            JNZ ESPERA
         IN AL, EST
            AND AL, 07FH
            OUT EST, AL
            INT 0
END
```

d) Modificar b) de modo que cuando se presione F10 se cancele la impresión. En tal caso, deben desactivarse las interrupciones del HANDSHAKE para evitar que se envíen más caracteres a imprimir.

```
IMR
      EQU 21H
EOI
      EQU 20H
      EQU 24H
INT0
INT2
      EQU 26H
HAND
      EQU 40H
EST
      EQU 41H
DATO EQU 40H
N_HND EQU 10
N_F10 EQU 15
        ORG 40
        DW RUT_HND
        ORG 60
        DW RUT_F10
        ORG 1000H
MSJ
        DB "UNIVERSIDAD NACIONAL DE LA PLATA"
FIN
        DB ?
TERMINO DB 0
        ORG 2000H
        MOV BX, OFFSET MSJ
        MOV CL, OFFSET FIN - OFFSET MSJ
        CLI
        MOV AL, ØFAH
        OUT IMR, AL
        MOV AL, N_F10
        OUT INTO, AL
        MOV AL, N_HND
        OUT INT2, AL
        MOV AL, 80H
        OUT EST, AL
        STI
 LAZO:
        CMP TERMINO, 0
        JZ LAZO
        INT 0
        ORG 3000H
RUT_HND: PUSH AX
         MOV AL, [BX]
         OUT DATO, AL
         INC BX
         DEC CL
         JNZ FINAL
         MOV AL, 0FFH
         OUT IMR, AL
         IN AL, EST
         AND AL, 7FH
         OUT EST, AL
```

```
MOV TERMINO, 1
  FINAL: MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
        ORG 3200H
RUT_F10: PUSH AX
         MOV AL, ØFFH
         OUT IMR, AL
         IN AL, EST
         AND AL, 7FH
         OUT EST, AL
         MOV TERMINO, 1
         MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
END
```

Parte 3: Ejercicios de repaso para el parcial

1) **Imprimir mensaje leído, longitud arbitraria** \uparrow . Escribir un programa que solicite ingresar caracteres por teclado y los almacene en la memoria. Al presionar la tecla F10, se envían a la impresora a través de la PIO. No es necesario mostrar los caracteres en la pantalla.

```
PA EQU 30h
PB EQU 31h
CA EQU 32h
CB EQU 33h
IMR EQU 21h
EOI EQU 20h
INTO EQU 24h
N_F10 EQU 10
ORG 40
DW RUT_F10
ORG 1000h
FIN_LEER DB 0
LARGO DB 0
CADENA DB ?
ORG 3200h
RUT_F10: PUSH AX
         MOV AL, 0FFh
         OUT IMR, AL
         MOV FIN_LEER, 1
         MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
ORG 3000h
LEE_CADENA: CLI
            MOV AL, 0FEh
            OUT IMR, AL
            MOV AL, N_F10
            OUT INTO, AL
            STI
            MOV BX, OFFSET CADENA
            MOV LARGO, 0
     SIGUE: INT 6
            INC BX
            INC LARGO
            CMP FIN_LEER, 0
            JZ SIGUE
            RET
ORG 2000h
     CALL LEE_CADENA
     MOV AL, OFDh; strobe salida (0), busy entrada (1)
     OUT CA, AL
     MOV AL, 0
                 ; puerto de datos todo salida
```

```
OUT CB, AL
     ; inicializo strobe en 0
     IN AL, PA
           AND AL, 11111101b
           OUT PA, AL
     MOV CL, LARGO
           MOV BX, OFFSET CADENA
           ; espero que busy=0
 POLL:
           IN AL, PA
           AND AL, 01h; 1000 0000
           JNZ POLL
           ; se que busy es 0, mandar caracer
           MOV AL, [BX]
           OUT PB, AL
           ; mandar flanco ascendente de strobe
           IN AL, PA
           OR AL, 00000010b
           OUT PA, AL
           ; strobe en 0
           IN AL, PA
           AND AL, 11111101b
           OUTPA, AL
           INC BX
           DEC CL
           JNZ POLL
           INT 0
END
```

```
PA EQU 30h
PB EQU 31h
CA EQU 32h
CB EQU 33h
IMR EQU 21h
EOI EQU 20h
INTO EQU 24h
N_F10 EQU 10
ORG 40
DW RUT_F10
ORG 1000h
ESTADO DB 0
ORG 3200h
RUT_F10: PUSH AX
         XOR ESTADO, 0FFh
         MOV AL, ESTADO
         OUT PB, AL
         MOV AL, EOI
```

```
OUT EOI, AL
POP AX
IRET

ORG 2000h
CLI
MOV AL, 0; puerto de datos todo salida
OUT CB, AL
OUT PB, AL
MOV AL, 0FEh
OUT IMR, AL
MOV AL, N_F10
OUT INT0, AL
STI
LOOP:JMP LOOP
END
```

Ruleta, F10 y azar * Escribir un programa que permita seleccionar un dígito al azar para jugar a la ruleta. Para eso, el programa principal debe iterar continuamente, cambiando un valor de un registro desde el '0' hasta el '9' (y volviendo al '0' luego del '9'). Cuando se presiona F10, la letra queda "seleccionada" en base al valor del registro, y debe mostrarse de inmediato en la pantalla. Luego el programa termina.

```
IMR EQU 21h
EOI EQU 20h
INTO EQU 24h
N F10 EQU 10
ORG 40
DW RUT_F10
ORG 1000h
FINALIZADO DB 0
NUM DB ?
ORG 3200h
RUT_F10: PUSH AX
         MOV FINALIZADO, 1
         MOV AL, 0FFh
         OUT IMR, AL
         MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
ORG 2000h
            CLI
            MOV AL, 0FEh
            OUT IMR, AL
            MOV AL, N_F10
            OUT INTO, AL
            STI
            MOV DL, '0'
LOOP:
            CMP FINALIZADO, 1
     JZ ELEGIDO
     INC DL
```

```
CMP DL, '9'

JNZ LOOP

MOV DL, '0'

JMP LOOP

ELEGIDO: MOV BX, OFFSET NUM

MOV [BX], DL

MOV AL, 1

INT 7

INT 0

END
```

4) **F10, lectura e Impresora con PIO** ★★★ Escribir un programa que debe mostrar en pantalla el mensaje "PRESIONE F10 PARA COMENZAR" y una vez que el usuario presiona F10, leer de teclado un mensaje de 10 caracteres. Este mensaje debe luego ser enviado a través de la pila a una subrutina para imprimirse en la impresora mediante el PIO. La configuración del PIO también debe hacerse en una subrutina aparte.

```
PA EQU 30h
PB EQU 31h
CA EQU 32h
CB EQU 33h
EOI EQU 20h
IMR EQU 21h
INTO EQU 24h
N_F10 EQU 10
ORG 40
DW RUT_F10
ORG 1000h
MSJ_EMPEZAR DB "PRESIONE F10 PARA COMENZAR"
FIN_MSJ_EMPEZAR DB ?
EMPEZAR DB 0
MENSAJE DB ?
ORG 3000h
CONFIG_PIO: MOV AL, 0FDh
            OUT CA, AL
            MOV AL, 0
            OUT CB, AL
            ; inicializo strobe en 0
            IN AL, PA
            AND AL, 11111101b
            OUT PA, AL
            RET
ORG 3200h
RUT F10:
          PUSH AX
          MOV EMPEZAR, 1
          MOV AL, EOI
          OUT EOI, AL
          POP AX
          IRET
```

```
ORG 3400h
LEER_CADENA: INT 6
             INC BX
             DEC CL
             JNZ LEER_CADENA
             RET
ORG 3600h
IMPRIMIR:
           MOV BX, SP
           ADD BX, 2
           MOV CL, [BX]
           ADD BX, 2
           MOV BX, [BX]
     POLL: IN AL, PA
           AND AL, 01h ; 1000 0000
           JNZ POLL
           MOV AL, [BX]
           OUT PB, AL
           ; mandar flanco ascendente de strobe
           IN AL, PA
           OR AL, 00000010b
           OUT PA, AL
           ; strobe en 0
           IN AL, PA
           AND AL, 11111101b
           OUT PA, AL
           INC BX
           DEC CL
           JNZ POLL
           RET
ORG 2000h
          MOV BX, OFFSET MSJ_EMPEZAR
          MOV AL, OFFSET FIN_MSJ_EMPEZAR - OFFSET MSJ_EMPEZAR
          INT 7
          CLI
          MOV AL, 0FEh
          OUT IMR, AL
          MOV AL, N_F10
          OUT INTO, AL
          STI
ESPERA:
          CMP EMPEZAR, 0
          JZ ESPERA
          CALL CONFIG_PIO
          MOV BX, OFFSET MENSAJE
          MOV CL, 10
          CALL LEER_CADENA
          MOV BX, OFFSET MENSAJE
```

```
PUSH BX
MOV CL, 10
PUSH CX
CALL IMPRIMIR
POP CX
POP AX
INT 0
```

```
PA EQU 30h
PB EQU 31h
CA EQU 32h
CB EQU 33h
ORG 1000h
ESTADO_LUCES DB ?
ORG 3000h
             MOV BX, SP
LUCES12:
             ADD BX, 2
             MOV BX, [BX]
             MOV AL, [BX]
             AND AL, 080h
             JZ FIN_LUCES12
             MOV AL, 0FFh
             OUT PB, AL
FIN_LUCES12: RET
ORG 2000h
         MOV AL, 0FFh
         OUT CA, AL
         MOV AL, 0
         OUT CB, AL
         IN AL, PA
         MOV ESTADO_LUCES, AL
         MOV AX, OFFSET ESTADO_LUCES
         PUSH AX
         CALL LUCES12
         POP AX
         INT 0
END
```

LEER_DIGITO muestra en pantalla el mensaje "INGRESE UN NUMERO DEL 1 AL 9:" y lee un carácter de teclado. Si el carácter ingresado no corresponde al número solicitado, se debe volver a leer un carácter hasta que el usuario ingrese efectivamente un número del 1 al 9.

DESCENDER recibe el dígito por valor y por la pila, y envía a la impresora en forma descendente desde el número ingresado hasta el "0", utilizando el Handshake por Consulta de Estado.

```
EOI
     EQU 20H
IMR
     EQU 21H
INTO EQU 24h
EST EQU 41H
DATO EQU 40H
N_F10 EQU 10
ORG 40
IP_F10 DW RUT_F10
ORG 1000H
CONTADOR
                 DB 39H
SEGUIR_ESPERANDO DB 0
                 DB "INGRESE UN NUMERO DEL 1 AL 9:"
MSJ_INGRESO
FIN_MSJ
                 DB ?
ORG 3200H
      RUT_F10: PUSH AX
               MOV SEGUIR_ESPERANDO, 0
               MOV AL, EOI
               OUT EOI, AL
               POP AX
               IRET
ORG 3000H
LEE_DIGITO: MOV BX, OFFSET MSJ_INGRESO
            MOV AL, OFFSET FIN_MSJ - OFFSET MSJ_INGRESO
            INT 7
            MOV BX, OFFSET CONTADOR
    REPITE: INT 6
            MOV DL, CONTADOR
            CMP DL, 31H
            JS REPITE
            CMP DL, 39H
            JNS REPITE
            RET
    ORG 3100H
    DESCENDER: MOV BX, SP
               ADD BX, 2
               MOV DL, [BX]
         POLL: IN AL, EST
               AND AL, 1
               JNZ POLL
               MOV AL, DL
               OUT DATO, AL
               INC BX
               DEC DL
               CMP DL, 2Fh
               JNZ POLL
```

```
RET
     ORG 2000H
         IN AL, EST
         AND AL, 7fh
         OUT EST, AL
         CALL LEE_DIGITO
         CLI
         MOV AL, 0FEH
         OUT IMR, AL
         MOV AL, N_F10
         OUT INTO, AL
         STI
         MOV SEGUIR_ESPERANDO, 1
 ESPERA: CMP SEGUIR_ESPERANDO, 1
         JZ ESPERA
         PUSH DX
         CALL DESCENDER
         POP DX
         INT 0
END
```

Éjemplo para enviar la cadena "ASDF": Envío de la "A" \rightarrow 5 segundos de espera \rightarrow Envío de la "S" \rightarrow 5 segundos de espera \rightarrow Envío de la "D" \rightarrow 5 segundos de espera \rightarrow Envío de la "F" \rightarrow 5 segundos de espera \rightarrow Envío de 0

```
PB EQU 31h
CB EQU 33h
EOI EQU 20h
IMR EQU 21h
INTO EQU 24h
INT1 EQU 25h
CONT EQU 10h
COMP EQU 11h
N_CLK EQU 10
N F10 EQU 15
ORG 40
DW RUT_CLK
ORG 60
DW RUT_F10
ORG 1000H
CADENA DB "Hola!"
FIN_CADENA DB ?
```

```
TIEMPO DB 0
TERMINAR DB 0
ORG 3000h
RUT_F10: PUSH AX
         MOV TERMINAR, 1
         MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
ORG 3200h
RUT_CLK: PUSH AX
         MOV TIEMPO, 1
         MOV AL, 0
         OUT CONT, AL
         MOV AL, EOI
         OUT EOI, AL
         POP AX
         IRET
ORG 3400h
ESPERAR:
          CLI
          MOV AL, 0FCh
          OUT IMR, AL
          MOV AL, 0
          OUT CONT, AL
          MOV TIEMPO, 0
          STI
          CMP TERMINAR, 1
LOOP:
          JZ SALIR
          CMP TIEMPO, 1
          JNZ LOOP
SALIR:
          CLI
          MOV AL, 0FEh
          OUT IMR, AL
          STI
          RET
ORG 2000h
    CLI
    MOV AL, 0FCh
    OUT IMR, AL
    MOV AL, N_F10
    OUT INT0, AL
    MOV AL, N_CLK
    OUT INT1, AL
    MOV AL, 0
    OUT CONT, AL
    MOV AL, 5
    OUT COMP, AL
    MOV AL, 0
    OUT CB, AL
    STI
    MOV BX, OFFSET CADENA
```

```
MOV CL, OFFSET FIN_CADENA - OFFSET CADENA

SIGUE: CMP TERMINAR, 1

JZ FINALIZAR

MOV AL, [BX]

OUT PB, AL

CALL ESPERAR

INC BX

DEC CL

JNZ SIGUE

FINALIZAR: MOV AL, 0FFh

OUT IMR, AL

INT 0

END
```